
11 Functions of More Than One Variable

11.1 Functions Given in Tabular Form

During the Winter you have probably heard weather reporters on radio or TV use the term wind chill factor, which
is a function that depends upon wind speed and air temperature. Table 1 illustrates that wind chill (how cold it
feels) in degrees Fahrenheit can be obtained as a function of wind speed in miles per hour and air temperature in
degrees Fahrenheit.

Air Temperature (◦F)
35 30 25 20 15 10 5 0 -5 -10

5 32 27 22 16 11 6 0 -5 -10 -15
10 22 16 10 3 -3 -9 -15 -22 -27 -34

Wind 15 16 9 2 -5 -11 -18 -25 -31 -38 -45
Speed 20 12 4 -3 -10 -17 -24 -31 -39 -46 -53
(mph) 25 8 1 -7 -15 -22 -29 -36 -44 -51 -59

30 6 -2 -10 -18 -25 -33 -41 -49 -56 -64
35 4 -4 -12 -20 -27 -35 -43 -52 -58 -67
40 3 -5 -13 -21 -29 -37 -45 -53 -60 -69
45 2 -6 -14 -22 -30 -38 -46 -54 -62 -70

Table 1: Wind Chill Table

To illustrate how Table 1 is used you may ask how cold does it feel if the wind is blowing at a speed of 15 miles
per hour and the air temperature is 20◦F. By moving horizontally along the row headed by 15 until you reach the
column headed by 20 you can see that the wind chill is −5◦F.

We have seen that functions of one variable may be represented in three different ways: in table form, graphi-
cally, or by means of a formula. When data is given in tabular form one can use the Maple V procedure surfdata
to plot a surface representing the function. This procedure is part of the Maple V library package plots.

> with(plots):

The procedure surfdata can now be invoked with the syntax:

> surfdata(L,<options>);

In the above the L represents a list of lists. We thus enter the data.

> L := [[[5,35,32],[10,35,22],[15,35,16],[20,35,12],[25,35,8],
> [30,35,6],[35,35,4],[40,35,3],[45,35,2]],[[5,30,27],[10,30,16],
> [15,30,9],[20,30,4],[25,30,1],[30,30,-2],[35,30,-4],[40,30,-5], [45,30,-
> 6]],[[5,25,22],[10,25,10], [15,25,2], [20,25,-3],[25,25,-7],[30,25,-
> 10],[35,25,-12],[40,25,-13],[45,25,-14]],[[5,20,16],[10,20,3],[15,20,-5],
> [20,20,-10],[25,20,-15],[30,20,-18],[35,20,-20],[40,20,-21],[45,20,-22]],
> [[5,15,11],[10,15,-3],[15,15,-11], [20,15,-17],[25,15,-22],[30,15,-25],
> [35,15,-27],[40,15,-29],[45,15,-30]],[[5,10,6],[10,10,-9],[15,10,-18],
> [20,10,-24],[25,10,-29],[30,10,-33],[35,10,-35],[40,10,-37],[45,10,-38]],
> [[5,5,0],[10,5,-15],[15,5,-25],[20,5,-31],[25,5,-36],[30,5,-41],[35,5,-
> 43],[40,5,-45],[45,5,-46]],[[5,0,-5],[10,0,-22],[15,0,-31],[20,0,-39],
> [25,0,-44],[30,0,-49],[35,0,-52],[40,0,-53] ,[45,0,-54]],[[5,-5,-10],[10,-
> 5,-27],[15,-5,-38],[20,-5,-46],[25,-5,-51], [30,-5,-56],[35,-5,-58],[40,-
> 5,-60], [45,-5,-62]],[[5,-10,-15],[10,-10,-34],[15,-10,-45],[20,-10,-
> 53],[25,-10,-59],[30,-10,-64],[35,-10,-67],[40,-10,-69],[45,-10,-70]]]:

Study the positioning of the brackets in the preceding Maple V statement, since the surfdata command is not
very friendly. Once the data from Table 1 has been entered the surface can be plotted. See Figure 5. There are a
number of options that can be used with surfdata, as well as any Maple V three dimensional plot routines. To get
a complete list of these options issue the command:

> ?plot3d[options]

The only options that will be used in this first Maple V segment are style, axes, and labels.

> P1 := surfdata(L):
> display(P1,style=PATCHCONTOUR,axes=FRAMED,labels=[W,T,WC]);

Figure 5: Wind Chill Factor Surface Plot

Figure 6: Contours of Wind Chill Temperatures

The option PATCHCONTOUR was one of seven style options and it caused the surface to have a shaded col-
oring along with contour or level curves. The option FRAMED we chose is one of four axes styles and causes
the figure to have the axes style that it has. The option labels instructed Maple V to label the coordinate axes, W
(wind speed), T (air temperature), WC, (wind chill). After you execute a Maple V three dimensional plot proce-
dure, a new window, which contains the plot, will appear on the sceen. Once you have the plot window you may
interactively change a number of the options within the plot. For example, by using the pull down menu associated
with the window you may change the style, color, axes, or projection options. The style menu has seven different
choices. Use the mouse button to select and choose the style option contour. The new plot, illustrating ten level
curves of wind chill temperatures, should look like Figure 6. If you want to write a Maple V command that will
give you Figure 6 directly then type in

> display(P1,style=CONTOUR,axes=FRAMED,labels=[W,T,WC]);

Figures 5 and 6 are representations on the plane of the page of a surface which lies in three space. Meteorolo-
gists use the term “extreme cold” when referring to wind chill temperatures below -20◦F. Figure 7 illustrates how
the wind chill surface compares with a plane parallel to the plane of wind speed and air temperature which is at
a constant wind chill factor of -20◦F. The following Maple V segment generates the horizontal plane WC = −20
and the wind chill surface on the same coordinate system.

> P2 := plot3d(-20,x=5..45,y=-10..35):
> display({P1,P2},style=PATCHCONTOUR,axes=FRAMED,labels=[W,T,WC]);

Figure 7: Extreme Cold below -20◦F

Figure 8: Contours of Wind Chill Temperatures

In Figure 7 temperatures below the plane represent wind chill temperatures that are in a state of extreme cold.
Sometimes it is a help in visualizing three dimensional objects to focus on level curves or contours. If you click the
mouse button when the cursor is in the window used for Figure 5, and select the style option contour then you will
get a surface like the one shown in Figure 8. Type in the following Maple V segment to obtain Figure 8 directly.

> display(P1,style=CONTOUR,axes=FRAMED,labels=[W,T,WC]);

Another way of representing this function graphically is to plot the data along contour curves on planar axes.
See Figure 9. This can be thought of as looking straight down the vertical axes (WC) at the surface. One way to
do this is to click the mouse button in the first plot window and turn framed box until the values of Theta and phi
are equal to 270 and 0 respectively. If you wish to issue a Maple V command that creates Figure 9 directly then
consider the following.

> display({P1},style=CONTOUR,orientation = [270,0],axes=FRAMED,
> labels = [W,C,WC]);

One problem with Figure 8 is that that we have not indicated the values of the level curves. We can accomplish this
by using the textplot3d command contained in the plots package. The syntax for this command is textplot3d( L,
options) where L is either a list with four components or a set consisting of such lists. The first three components
of the list are the coordinates in three space where you wish to have text written on your three dimensional plot
and the fourth component is a string with the text that you want written. The next Maple V segment shows how
to use textplot3d to label some of the level curves. See figure Figure 9.

> T1 := textplot3d([5,30,27,‘27‘],color=BLACK):
> T2 := textplot3d([5,20,16,‘16‘],color=BLACK):
> T3 := textplot3d([25,30,1,‘1‘],color=BLACK):
> T4 := textplot3d([35,20,-20,‘-20‘],color=BLACK):
> display({P1,T1,T2,T3,T4},orientation =[270,0],style=CONTOUR,axes=FRAMED,
> labels = [W,T,‘‘],view = [0..50,-15..40,-80..40]);

In the preceding command we introduced the option view in order to give some empty space between the graph
and axes. There are other subjective terms used to describe wind chill temperature: if the wind chill is below 32◦F
then the weather is said to be cold, when below about 15◦F very cold, when below 0◦F bitter cold, and when below
-20◦F extreme cold. See Figure 10.

> T5 := textplot3d([8,35,32,‘COLD‘],color =BLACK):

Figure 9: Labeled Contour Curves

Figure 10: Contours Descriptive Text Added

> T6 := textplot3d([14,25,0,‘VERY COLD‘],color =BLACK):
> T7 := textplot3d([20,20,0,‘BITTER COLD‘],color =BLACK):
> T8 := textplot3d([28,10,0,‘EXTREME COLD‘],color =BLACK):
> display({P1,T1,T2,T3,T4,T5,T6,T7,T8},view = [0..50,-15..40,-80..40],
> style=contour,orientation = [270,0],axes=FRAMED);

In the next graph we use the plot of the plane WC = -20 to shade the region where extreme cold occurs. See
Figure 11

> display({P1,P2,T1,T2,T3,T4},orientation=[270,0],view = [0..50,-15..40,
> -80..40],style=PATCHCONTOUR,axes=FRAMED);

16 -20

27 1

0 10 20 30 40 50

-10

0

10

20

30

40

Figure 11: Contours with Region of Bitter Cold Shaded

Exercises 11.1

1. Table 2 shows the heat index as a function of temperature and humidity. The heat index is the temperature
which tells you how hot it feels to you for a given air temperature and humidity.

Air Temperature (◦F)
70 75 80 85 90 95 100 105 110 115

0 64 69 73 78 83 87 91 95 99 103
10 65 70 75 80 85 90 95 100 105 111

Relative 20 66 72 77 82 87 93 99 105 112 120
humidity 30 67 73 78 84 90 96 104 113 123 135

(%) 40 68 74 79 86 93 101 110 123 137 151
50 69 75 81 88 96 107 120 135 150 165
60 70 76 82 90 100 114 132 149 167 187

Table 2: Heat Index

(a) If the temperature is 90◦F and the humidty is 50%, how hot does it seem to be?

(b) Estimate what the relative humidity should be for 100◦F to feel like 100◦F.

(c) Make a lists of lists in a Maple V session and use surfplots to plot a graph of this data.

(d) Heat exhaustion is likely to occur when the heat index rises to 105◦F. Make a Maple V plot showing
contour lines of constant heat index in the relative humidity/temperature plane and shade out the region
in which heat exhaustion is likely to occur.

11.2 Functions of More Than One Variable Given by Formulas

As an example of some of the things that can be done when studying functions of two variables define a function
z = f (x, y) as follows:

z = sin(x) cos(y).

As with a function of one variable you can enter it as an expression:

>f := sin(x)*cos(y);

f := sin( x ) cos( y )

The subs command can be used to find the value of the expression at, say, the point (π/4, π/3).

> subs({x=Pi/4,y=Pi/3},f);

sin

(
1
4

π

)
cos

(
1
3

π

)
Perhaps this answer is better if written as follows:

> eval(");

1
4

√
2

It is sometimes useful to obtain the value of the function in floating point form. In this case the command evalf
is used.

> evalf(");

.3535533905

If it is desired that the expression f be given in terms of a function, then one can define the function directly
with the statement

> f := (x,y)-> sin(x)*cos(y);

or apply the command unapply.

> h := unapply(f,x,y);

h := ( x, y ) → sin( x ) cos( y )

In either case evaluating f at (π/4, π/3) is now accomplished as follows:

> h(Pi/4,Pi/3);

1
4

√
2

> evalf(");

Figure 12: Plot of z = sin x cos y

.3535533905

In Maple V there are two ways to study a function: as an expression or as a Maple defined function. Which do
you prefer?

The graph of a function of the form z= f(x,y) is a surface. The graph of the expression f for

−π ≤ x ≤ π, −π ≤ y ≤ π

is shown in Figure 12 and can be obtained with the command plot3d as follows:

> plot3d(f,x=-Pi..Pi,y=-Pi..Pi);

Recall that there are differences in syntax when plotting an expression, as “f” is in this example, and a function,
as “h” is in this example, when using the plot command for the one variable case. The same plot,Figure 12, as above
can be also obtained for the function h with the command plot3d.

> plot3d(h(x,y),x=-Pi..Pi,y=-Pi..Pi);

If the plot3d command is properly executed then Maple V will open a new window with the 3d plot included
in it. You can view this plot from different angles as was done in the previous section. Do this by clicking on the
plot with the left mouse button and holding the button down. A box will appear in place of the plot. Move the
mouse while holding the left mouse button down and you view the box from different angles. The Theta and Phi
in the upper left corner describe the view angles. Let go of the left mouse button. If you wish Maple to show the
plot with the new angle then press and release the middle mouse button.

There are also many other options available in the 3d plot window. The style, color, axes, or projection can
be changed through pull down menus. The plot can be copied and pasted into the the worksheet in the same way
as was demonstrated in the introductory lesson.
HINT: You can make your 3d plot window smaller and paste the result into the worksheet to reduce the amount
of paper needed when printing hard copy versions.
WARNING: It is important to keep in mind that the syntax for plotting Maple V expressions is different than the
one for plotting Maple V functions. The following two Maple commands lead to an error messages and do not
provide plots.



          

11 FUNCTIONS OF MORE THAN ONE VARIABLE 19

> plot3d(f,-Pi..Pi,-Pi..Pi); # Here f is an expression
# you should specify the ranges

# with x= -Pi..Pi, y=-Pi..Pi

> plot3d(h(x,y),-Pi..Pi,-Pi..Pi); # Here h is a function
# you should specify the ranges

# with x= -Pi..Pi, y=-Pi..Pi

Nevertheless, the following command is okay.

> plot3d(h,-Pi..Pi,-Pi..Pi);

If you wish to make a multiple plot involving two surfaces, then this can be accomplished in the same way that
multiple plots can be made with curves. We now plot the function h(x, y) defined above along with the expression
x2 sin(y). See Figure 13.

> plot3d({x^2*sin(y),h(x,y)},x=-Pi..Pi,y=-Pi..Pi);

Figure 13: Plot of z = sin x cos y and z = x2 sin y

The set of Maple V library routines that reside in plots contains many procedures that help in analyzing func-
tions of two or more variables. In particular the command contourplot can be used to plots level curves of a func-
tion of two variables. See Figure 14.

> with(plots):
> contourplot(h(x,y),x=-Pi..Pi,y=-Pi..Pi);

You can move the plot in this window around to view this plot from different angles.
To get an idea about the relationship between contourplot and plot3d study the plot resulting from the follow-

ing Maple V command.

> plot3d(h(x,y),x=-Pi..Pi,y=-Pi..Pi,style=contour,
orientation = [270,0]);

Figure 14: A Contourplot of z = sin x cos x

Indeed using the option style = contour as in the last command gives exactly the same plot, Figure 14, as con-
tourplot . Try it. The above Maple V segment shows that contours are easy to obtain but the previous commands
do not provide much insight into which level a particular curve corresponds.

Let us now illustrate how to use Maple V to obtain the level curve h(x, y) = 0.6. First we make a plot of the
surface z = h(x, y) and label it “P1”.

> P1 := plot3d(h(x,y),x=-Pi..Pi,y=-Pi..Pi,orientation = [15,60]):

Using the command implicitplot3d (from plots) we plot the plane h = 0.6.

> P2 := implicitplot3d(z=0.6,x=-Pi..Pi,y=-Pi..Pi,z= -1..1,
style=patchnogrid):

Now we display the surface z = h(x, y) along with the plane z = 0.6. Observe that the plane intersects the
surface in three curves (level curves). See Figure 15.

> display3d({P1,P2},tickmarks=[4,4,4],orientation = [15,60],axes=boxed);

You can use the plots Library routine implicitplot to obtain the level curves or contour

Figure 15: The Plane z = 0.6 and the Surface z =
sin x cos y

curves for z = h(x, y) at z = 0.6. These curves represent the projection onto the xy-plane of the curve cut out by
z = 0.6 from z = h(x, y). See Figure 16.

> implicitplot(h(x,y) = 0.6, x=-Pi..Pi, y=-Pi..Pi);

Figure 17: Level Curves for sin x cos y = 0.6 Figure 18: Contours for sin x cos y = 0.6

In the following the plots procedure implicitplot3d is used to illustrate that the level curves are also represented
as the intersection of three cylinders defined implicitly in three space by

h(x, y) = 0.6.

See Figure 17.

> implicitplot3d({h(x,y) = 0.6, z=0.6} , x = -Pi..Pi, y=-Pi..Pi, z=-1..1);

Figure 18 shows the same curves after you have changed the orientation so that θ = 270, and φ = 0. The fol-
lowing Maple V command also creates this plot.

> implicitplot3d({h(x,y) = 0.6, z=0.6} , x = -Pi..Pi, y=-Pi..Pi, z=-
> 1..1,orientation=[270,0],style=PATCHCONTOUR);

The next example deals with a simpler function,

j(x, y) = x2 + y2,

that permits more explicit representations. First define j(x,y).

> j := (x,y) -> x^2+y^2;

The graph of this function over the square domain:

−3 ≤ x ≤ 3, −3 ≤ y ≤ 3

is shown in Figure 19.

> plot3d(j(x,y),x=-3..3,y=-3..3);

Or we can plot the graph over the over the circular domain:

x2 + y2 ≤ 9.

See Figure 20.

> P1 := plot3d(j(x,y), x=-3..3, y= -sqrt(9-x^2)..sqrt(9-x^2)): ";

Figure 19: Plot of z = x2 + y2 over a Square Region Figure 20: Plot of z = x2 + y2 over a Disk

The plane z = 6 is plotted next.

> P2 := plot3d(6,x=-3..3, y= -3..3,style = patchnogrid): ";

> plot3d(j(x,y),x=-3..3,y=-3..3);

The curve cut from the surface z = j(x,y) by the plane z = 6 is shown in Figure 21. The following Maple V

Figure 21: Plot of z = x2 + y2 and Plane z = 6. Figure 22: Curve of Intersection of z = x2 + y2 and
Plane z = 6.

command produces it.

> display3d({P1,P2},axes=frame, tickmarks = [4,4,4],orientation=[45,60]);

In order to get a better idea of this intersection curve observe that any point with coordinates

(
√

6 cos(t),
√

6 sin(t), 6)

lies on both the plane z = 6 and the surface z = j(x, y). This means that the curve of intersection of the surface
and the plane is parameterized by the following three equations:

x =
√

6 cos(t), y =
√

6 sin(t), z = 6.

Using the plots package procedure spacecurve we obtain Figure 22 as follows:

> P3 := spacecurve([sqrt(6)*cos(t), sqrt(6)*sin(t),6],t=0..2*Pi,color=black,
> thickness=3): ";

Figure 23 shows curve of intersection along with the two surfaces.

> display3d({P1,P2,P3},tickmarks=[4,4,4],orientation = [40,120],axes=boxed);

Figure 23: Plot of z = x2 + y2, z = 6 and the Curve of
Intersection

Figure 24: Curve of Intersection Projected onto xy-
Plane

The contour curves are the level curves projected onto the z = 0 plane, and thus the following Maple V com-
mand creates a plot of this curve. See Figure 24.

> P4 := spacecurve([sqrt(6)*cos(t),sqrt(6)*sin(t),0],t=0..2*Pi,color=red,
> thickness=3):

A three dimensional plot of the curve of intersection of z = 6 and z = j(x, y), the surfaces themselves and the
contour curve on the same figure is given below.

> display3d({P1,P2,P3,P4},tickmarks=[4,4,4],orientation =
> [40,120],axes=boxed);

The contour curve is the projection of the curve of intersection of the plane z = 6 and the surface z = j(x, y)

onto the xy-plane. These two curves lie on the cylinder j(x, y) = 6.

> P5 := implicitplot3d(j(x,y)=6,x=-3..3,y=-3..3,z= 0..6): ";

The following plot shows all of this. See Figure 26.

> display3d({P1,P2,P3,P4,P5},tickmarks=[4,4,4],orientation =
> [40,120],axes=boxed);

Now we illustrate how to graph certain curves that lie on the surface

z = x2 + y2.

First the surface is plotted using the built in Maple V command plot3d.

> P1 := plot3d(j(x,y),x=-3..3,y=-3..3): ";

We now wish to show the plane x = 2 cutting the surface

z = x2 + y2.

To do this we use the built in Maple V library routine in plots called implicitplot3d to plot the plane.

> P2 := implicitplot3d(x=2,x=-3..3,y=-3..3,z=0..20,style=patchnogrid):";

Figure 25: Cylinder x2 + y2 = 6.

Figure 26: All of the Above

The Maple V library routine in plots called display3d is used to plot both the plane and the surface

z = x2 + y2.

which is shown in Figure 27.

> display3d({P1,P2},tickmarks=[4,4,4],orientation = [40,120], axes=framed);

Figure 27: Intersection of Plane x = 2 with z = x2 + y2 Figure 28: Curve of Intersection of Plane x = 2 and z =
x2 + y2

The plane x = 2 and the surface
z = j(x, y) = x2 + y2

intersect in a curve. This curve is represented parametrically by the equations:

x = 2, y = t, z = j(2, t) = 22 + t2.

You should make sure that you understand why all the points (x, y, z) that satisfy the above three equations
also satisfy both the equation z = j(x, y) and x = 2.

The Maple V library procedure in plots called spacecurve is now used to plot this curve. See Figure 28.

> P3 := spacecurve([2,t,j(2,t)],t=-3..3,thickness=3,color=black): ";

Now the combined plot of the surface, the plane, and the curve of intersection is obtained. See Figure 29.

> display3d({P1,P2,P3},tickmarks=[4,4,4],axes=framed,orientation=[40,120]);

Figure 29: Intersection of Plane x = 2, z = x2 + y2

with Curve of Intersection
Figure 30: Curve of Intersection of Plane y = 3 and
z = x2 + y2

Similarly other curves that lie on the intersection of the surface and a plane parallel to the xz- plane are plotted.
The curve of intersection of the plane y = 3 and the surface z = j(x, y) is represented parametrically by

x = t, y = 3, z = j(t, 3).

This curve is shown in Figure 30 and is plotted as follows:

> P4 := spacecurve([t,3,j(t,3)],t=-3..3,thickness=3,color=black):";

Finally, the curve which the plane y = −2 cuts out of the surface is plotted as follows:

> P5 := spacecurve([t,-2,j(t,-2)],t=-3..3,thickness=3,color=black):";

The three curves and the surface are plotted together in Figure 31 by the next Maple V command.Can you
identify each of the curves in the plot with the corresponding plane?

> display3d({P1,P3,P4,P5},tickmarks=[4,4,4],axes=framed,style = wireframe,
> labels = [x,y,z]);

Exercises 11.2

1. Plot the portion of the plane
2x + y + 3z = 6

which lies in the first octant along with the line segments of intersection of the plane with the coordinate
planes.

2. Plot the sphere of radius 4 centered at the origin, along with the latitudinal circle of intersection of the sphere
with the plane z = 1.

3. Use the parametrization

x = (4 + 2 sin u) cos v, y = (4 + 2 sin u) sin v, z = 4 + 2 cos u, 0 ≤ u ≤ 2π, 0 ≤ v ≤ 2π

to make a Maple V plot for the torus, created by rotating the circle of radius 2 centered at the point (4,0,0)
about the z-axis.

Figure 31: Plot of Three Curves and Surface z = x2 + y2

11.3 Linear Functions

Consider an equation of the form:
z = c + mx + ny.

The graph of this equation is a plane and intersects the z- axis at the point (0,0,-c). If m is not 0 the plane inter-
sects the the x-axis at the point (−c/m, 0, 0). If n is not 0 the plane intersects the y-axis at the point (0,−c/n, 0).

What can be said in the event that both m or n are missing ?

As an example we will now illustrate how to use Maple V to plot the plane:

z = 5 − 2x − y.

> P1 := plot3d(5-2*x-y,x=-5..5,y=-5..5,style=patchnogrid):";

In order to have a better understanding of the graph of this plane, it is sometimes helpful to plot the intersection
of the plane with the coordinate planes. The intersection of the plane x = 0 and the plane z = 5 − 2x − y has a
parametric representation:

x = 0, y = t, z = 5 − t.

Show this. Hint: Observe that for any t the point with coordinates (0, t, 5 − t) lies on both planes.
The next Maple V segment indicates how to plot a segment of this line for 0 ≤ t ≤ 5 using the Maple V pro-

cedure spacecurve from the library package plots.

> with(plots):
> P2 := spacecurve([0,t,5-t],t=0..5,color=black):";

Similarly, the next two Maple commands plot the traces in the planes y = 0, z = 0 respectively.

> P3 := spacecurve([t,0,5-2*t],t=0..5/2,color=black):";

> P4 := spacecurve([5/2-t/2,t,0],t=0..5,color=black):";

Figure 32 shows a plot of all of this.

> display3d({P1,P2,P3,P4},tickmarks=[4,4,4],axes=normal);

Figure 32: The planes z = 5 − 2x − y and Traces with
the Coordinate Planes

-400

-200

0

200

400

-400

-200

0

200

400

-1000

0

1000

Figure 33: Plane through the Points
(100, 0, 102), (101,−100, 303), and (305, 0,−105).

Example 11.3.1 Find the equation of the plane that passes through the three points (100, 0, 102), (101,−100, 303),

and (305, 0,−105). Make a plot of this plane.

Solution The first part of this problem can be solved by substituting the coordinates of the three points into z = c
+ mx + ny and then solving the resulting three equations for c, m, and n using solve.

> eq1 := subs({x=100,y=0,z=102},z=c+m*x+n*y);

eq1 := 102 = c + 100m

> eq2 := subs({x=101,y=-100,z=303},z=c+m*x+n*y);

eq2 := 303 = c + 101m − 100n

> eq3 := subs({x=305,y=0,z=-105},z=c+m*x+n*y);

eq3 := −105 = c + 305m

> sol := solve({eq1,eq2,eq3},{c,m,n});

sol :=
{

n = −10353
5125

, m = −207
205

, c = 8322
41

}
Substituting this result into the general equation gives the desired equation.

> eqn := subs(sol,z=c+m*x+n*y);

eqn := z = 8322
41

− 207 x
205

− 10353 y
5125

The graph of this plane can be plotted using either plot3d or implicitplot3d. See Figure 33

> implicitplot3d(eqn,x=-500..500,y=-500..500,z=-1500..1500,style=patchnogrid,
> axes=boxed, tickmarks = [4,4,4] );

WARNING: You can’t plot an equation using plot3d.

> plot3d(eqn,x=-5..5,y=-5..5,style=patchnogrid,axes=boxed,
> tickmarks = [4,4,4]);
Error, (in plot3d) invalid 1st argument (the function) , z = 8322/41-207/205*
x-10353/5125*y

But you can plot an expression. In this problem the right hand side of the equation can be plotted using the
Maple command rhs.

> plot3d(rhs(eqn),x=-5..5,y=-5..5,style=patchnogrid,axes=boxed,
> tickmarks = [4,4,4]);

Example 11.3.2 Determine whether or not the points either of following sets of points lie on a plane.

SetA = {(100, 0, 102), (101,−100, 303), (305, 0,−105), (1025, 364,
80232617

5125
)},

and

SetB = {(100, 0, 102), (101,−100, 303), (305, 0,−105), (1025, 364,−80232617
5125

)},

Solution The first three points in each set lie on the same plane as in the preceding example. Thus if the fourth
point in each set satisfies the equation for the plane found in that example then it lies on the plane.

> subs({x=1025,y=364,z=80232617/5125},eqn);

8032617
5125

= −8032617
5125

Thus SetA is not co-planar.

> subs({x=1025,y=364,z=-8032617/5125},eqn);

−8032617
5125

= −8032617
5125

We conclude that all of the points in set SetB lie on a plane.

Exercises 11.3

1. Find the equation of the plane determined by the three points A : (1, 2,−1), B : (2, 3, 1), and C : (3,−1, 2).

2. Show that the z intercept of the plane through the points A : (100,−200, 300), B : (150, 300, 200), C :
(150, 300, 200), and C : (125, 150, 100).

11.4 Functions of More Than Two Variables

The graph of a function of two variables is a surface or three dimensional object. The graph of a function of three
variables is a three dimensional solid.

As an example consider the function

f (x, y, z) = x2 + y2 − z2.

One way to obtain geometric insight into a surface is to study level curves. Analogously you can gain geometric
insight into the graphs of functions of three variables by plotting level surfaces. Consider the level surfaces:

f (x, y, z) = 1, f (x, y, z) = −1, f (x, y, z) = 0.

Solving for z you can find explicit formulas, in this very special case, for each of the level surfaces:
If f (x, y, z) = 1 then z =

√
x2 + y2 − 1 or z = −

√
x2 + y2 − 1. If f (x, y, z) = −1, then z =

√
x2 + y2 + 1 or

z= -
√

x2 + y2 + 1. If f(x,y,z)=0,then z =
√

x2 + y2 or z = −
√

x2 + y2.

In the first case f (x, y, z) = 1 we get a hyperboloid of one sheet. We now we use implicitplot3d which is
contained in the plots library package. See Figure 34.

> with(plots):
> implicitplot3d(x^2+y^2-z^2= 1,x=-3..3,y=-3..3,z=-5..5,style=patchcontour,
> axes=framed,tickmarks=[3,3,5],orientation = [40,80]);

Figure 34: Level Surface x2 + y2 − z2 = 1 Using im-
plicitplot3d

Figure 35: Level Surface x2 + y2 − z2 = 1 Using Pa-
rameterizations

The command implicitplot3d gives an acceptable plot in this case, but if we parameterize the equation we can
get a better plot. Observe that for any pair of numbers (r, θ) the point

(r cos(θ), r sin(θ),
√

r2 − 1),

satisfies the equation
z =

√
x2 + y2 − 1

and hence lies on the hyperboloid for points in which the z-coordinate is non-negative. Similarly, the point

(r cos(θ), r sin(θ),−
√

r2 − 1)

is on the hyperboloid for z non-positive. The following plot is thus of the top half.

> P1 := plot3d([r*cos(theta),r*sin(theta),sqrt(r^2-1)],r=0..3,
> theta=0..2*Pi): ";

Next we obtain the bottom half.

> P2 := plot3d([r*cos(theta),r*sin(theta),-sqrt(r^2-1)],r=0..3,
> theta=0..2*Pi): ";

Both halves together. See Figure XIfig31

> display3d({P1,P2},style=patchcontour,orientation=[45,75],
> axes=framed,tickmarks = [3,3,5]);

When x2 + y2 − z2 = 0, the graph is a cone. The plot using implicitplot3d is too crude in that it does not look
like a cone. See Figure 36.

> implicitplot3d(x^2+y^2-z^2,x=-3..3,y=-3..3,z=-5..5,style=patchcontour,
> axes=framed,tickmarks=[3,3,5],orientation = [40,80]);

Figure 36: Level Surface x2 + y2 − z2 = 0 Using im-
plicitplot3d

Figure 37: Level Surface x2 + y2 − z2 = 0 Using Pa-
rameterizations

Parameterizing we get the upper and lower halves of the cone. See Figure 37.

> P1 := plot3d([r*cos(theta),r*sin(theta),sqrt(r^2)],r=0..3,
> theta=0..2*Pi): ";
> P2 := plot3d([r*cos(theta),r*sin(theta),-sqrt(r^2)],r=0..3,
> theta=0..2*Pi): ";
> display3d({P1,P2},style=patchcontour,orientation=[45,75],
> axes=framed,tickmarks = [3,3,5]);

Finally, the graph of the equation x2 + y2 − z2 = −1 is a hyperboloid of two sheets. See Figure 38.

> implicitplot3d(x^2+y^2-z^2=-1,x=-3..3,y=-3..3,z=-5..5,style=patchcontour,
> axes=framed,tickmarks=[3,3,5],orientation = [40,80]);

Parameterizing gives a better view. See Figure 39

> P1 := plot3d([r*cos(theta),r*sin(theta),sqrt(r^2+1)],r=0..3,
> theta=0..2*Pi): ";
> P2 := plot3d([r*cos(theta),r*sin(theta),-sqrt(r^2+1)],r=0..3,
> theta=0..2*Pi): ";
> display3d({P1,P2},style=patchcontour,orientation=[45,75],
> axes=framed,tickmarks = [3,3,5]): ";

Example 11.4.1 Use implicitplot3d to plot the Catenoid

cosh z =
√

x2 + y2.

Figure 38: Level Surface x2 + y2 − z2 = −1 Using im-
plicitplot3d

Figure 39: Level Surface x2 + y2 − z2 = −1 Using Pa-
rameterizations

11.5 Vectors

When studying vectors using Maple V it is useful to employ procedures that are part of the Maple V Linear Algebra
package called linalg. This package is made available by the with(linalg) command.

> with(linalg):
Warning: new definition for norm

Warning: new definition for trace

Ignore the warnings, they will not cause you any trouble in this lesson.
Enter the following vectors

A =< 1, 2, 3 >, B =< 5, 0, 5 >, C =< x, y, z >

in order to illustrate some of the basic Maple V commands that deal with vectors. (Note: Throughout this book
we will use the notation < a, b, c > for the vector

a
−→
i + b

−→
j + c

−→
k .)

The command vector is available to your session once the command with(linalg) has been entered. There are
several equivalent syntaxes for this command. We will use

> vector([x1, · · · , xn]);

as the command to enter the vector < x1, · · · , xn > . Now enter the vectors A, and B.

> A := vector([1,2,3]);

A := [ 1 2 3 ]

> B := vector([5,0,5]);

B := [ 5 0 5 ]

If you want to have Maple V print out the value of a vector you will need to use evalm, evaluate to a matrix.
For example, the command

> A;

A

does not show you the values of the components of the vector A, but the commands

> evalm(A);

[ 1 2 3 ]

> print(A);

[ 1 2 3 ]

do.
The vector C has variable components and is entered in the same way:

> C := vector([x,y,z]);

C := [ x y z ]

Assigning values to the variables in a vector can prove a little tricky. For example.

> evalm(subs({x=1,y=2,z=3},C));

[ x y z ]

The following is a rather straight forward method for making a substitution into a vector.

> evalm([seq(subs({x=1,y=2,z=3},C[i]),i=1..3)]);

[ 1 2 3 ]

On the other hand the procedure map is more elegant:

> map(t ->subs({x=1,y=2,z=3},t),C);

[ 1 2 3 ]

In an active Maple V session you can’t visually distinguish a vector from a list. To make such distinctions use
the Maple V command type.

> type(",vector);

true

The fact that the Maple V structure vector differs from a list can be illustrated as follows:

> type([1,2,3],vector);

false

But, the linalg procedure evalm, evaluate matrix, converts a list to a vector.

> type(evalm([1,2,3]),vector);

true

> type(A,vector);

true

Observe that even though a vector and a list appear to you to be the same, Maple V regards them as differ-
ent objects. A vector must be defined to the session with the command vector. This takes an argument which is
enclosed within the ”[ ... ]”.

We can add vectors. There are two ways that this can be accomplished. One way is to use the linear algebra
routine add.

> S := add(A,B);

S := [ 6 2 8 ]

This looks like and is a vector.

> type(S,vector);

true

Remember that to show the components of a vector on the screen we have to use evalm.

> evalm(S);

[ 6 2 8 ]

Otherwise we get the unevaluated variable:

> S;

S

Note that you can use the add command to add lists that look like vectors, and the result is a vector:

> add([1,2,3],[5,0,5]);

[ 6 2 8 ]

> type(",vector);

true

The following is another way to add vectors:

> evalm(A+B);

[ 6 2 8 ]

> type(",vector);

true

Also, one can add two lists and use evalm to convert the result to a vector.

> evalm([1,2,3]+[5,0,5]);

[ 6 2 8 ]

> type(",vector);

true

There are several ways to calculate the length of a vector with Maple V. Recall that if V = < a, b, c > then the
length of V, ||V ||, is

||V || =
√

a2 + b2 + c2.

Thus the most natural way to obtain the length of the vector A, defined above, is as follows:

> i := ’i’; # This resets "i" so that it can be used as an index.

i := i

> sqrt(sum(A[i]^2,i=1..3));

√
14

Or if you want this result in floating point decimal to 15 digits:

> evalf(",15);

3.74165738677394

> sqrt(sum(C[i]^2,i=1..3)); √
x2 + y2 + z2

Within the linalg package is a built-in procedure norm which can be used to compute the length:

> norm(A,2);

√
14

> norm(C,2); √
|x|2 + |y|2 + |z|2

Make sure that you use the option 2 when using norm otherwise you will not get the standard length.
The vector operations dot product and cross product are available in linalg. Their syntax and output are obvi-

ous, if you know the mathematical meaning of the terms.

> dotprod(A,B);

20

> dotprod(B,A);

20

> dotprod([5,0,5],[1,2,3]);

20

> dotprod(A,C);

x + 2 y + 3 z

This allows us to calculate the length by yet another method.

> sqrt(dotprod(A,A));

√
14

> LengthOfC := sqrt(dotprod(C,C));

LengthOf C :=
√

x2 + y2 + z2

The procedure subs can be used to assign values when the length is a variable:

> subs({x=3,y=4,z=5},LengthOfC);

√
50

Now for some examples using crossprod.

> crossprod(A,B);

[ 10 10 − 10 ]

> crossprod(B,A);

[−10 − 10 10 ]

> crossprod([1,2,3],[5,0,5]);

[ 10 10 − 10 ]

> crossprod(A,C);

[ 2 z − 3 y 3 x − z y − 2 x ]

One can also perform scalar multiplication using the procedure scalarmul from linalg.

> scalarmul(A,1/10); [
1

10
1
5

3
10

]
> scalarmul([1,2,3],1/10); [

1
10

1
5

3
10

]
> type(",vector);

true

> evalm(1/10*A); [
1

10
1
5

3
10

]
> evalm(1/10*[1,2,3]); [

1
10

1
5

3
10

]
One can use scalarmul to normalize a vector:

> U := scalarmul(A,1/sqrt(dotprod(A,A)));

U :=
[

1
14

√
14

1
7

√
14

3
14

√
14

]
> sqrt(dotprod(U,U));



            
11 FUNCTIONS OF MORE THAN ONE VARIABLE 38

1

Also the linalg package has its own normalization procedure normalize.

> normalize(A); [
1

14

√
14

1
7

√
14

3
14

√
14

]
Example 11.5.1: Let F = < −1, 1, 3 > and G = < 2, 4, 3 > be vectors. Find vectors Fparallel and Fperp such that
F = Fparallel + Fperp, and such that Fparallel is parallel to G and Fperp is perpendicular to G.

Solution: The first step is to define the vectors F and G.

> F := vector([-1,1,3]); G := vector([2,4,3]);

F := [−1 1 3 ]

G := [ 2 4 3 ]

The vector Fparallel is obtained by taking the component of F in the direction of G.

> Fparallel := evalm(G*dotprod(F,G)/dotprod(G,G));

Fparallel :=
[

22
29

44
29

33
29

]
The vector Fperp is the difference between F and Fparallel .

> Fperp := evalm(F-Fparallel);

Fperp :=
[−51

29
−15
29

54
29

]
As a check.

> evalm(F) = evalm(Fparallel+Fperp);

[−1 1 3 ] = [−1 1 3 ]

Thus F = Fparallel + Fperp,

> dotprod(G,Fperp);

0

> crossprod(G,Fparallel);

[ 0 0 0 ]

Now do you know for sure that Fparallel and Fperp are correct?

Example 11.5.2: Find the equation of the plane though the point (1, 1, 2) which is perpendicular to the vector
< 3, 5, 2 >.

Solution: For the point P with coordinates (x, y, z) to lie on the required plane the vector < x − 1, y − 1, z − 2 >

should be perpendicular to the vector < 3, 5, 2 > . Hence

> dotprod([3,5,2],[x-1,y-1,z-2]) = 0;

3 x − 12 + 5 y + 2 z = 0

We can show how to find the equation of a plane through three points using vector methods.
Example 11.5.3: Find the equation of the plane that passes through the three points P = (1, 0, 1), Q =

(1,−1, 3), and R = (3, 0,−1).

Solution: Since P and Q lie in the plane the vectors U and V are also in the plane where

> U := evalm([1,-1,3]-[1,0,1]); V := evalm([3,0,-1]-[1,0,1]);

U := [ 0 − 1 2 ]

V := [ 2 0 − 2 ]

A vector which is perpendicular to the plane is thus obtained by taking the cross product of U and V.

> N := crossprod(U,V);

N := [ 2 4 2 ]

The equation of the plane is thus:

> dotprod(N,[x-1,y,z-1])=0;

2 x − 4 + 4 y + 2 z = 0

Example 11.5.4: Find the area of the triangle formed by the points with vertices:

A = (−122, 317, 615), B = (217, 314, 617), and C = (3117,−217, 615).

Solution: The vectors AB, and AC are given by:

> AB := linalg[vector]([217-(-122),314-317,617-615]);

AB := [ 339 − 3 2 ]

> AC := linalg[vector]([3117-(-122),-217-317,615-615]);

AC := [ 3239 − 534 0 ]

The area of the parallelogram determined by the vectors AB and AC is equal to the length of the crossproduct

AB × AC.

The area of the triangle is

> AreaTriangle := linalg[norm](linalg[crossprod](AB,AC),2)/2;

AreaTriangle := 1
2

√
29389878589

Example 11.5.5: Find the volume of the parallopiped formed by the vectors

A =< 1, 2, 3 >, B =< −1, 0, 1 >, andC =< 2, 5, 5 > .

Solution:

> Area := abs(dotprod([1,2,3],crossprod([-1,0,1],[2,5,5])));

Area := 6

Exercise 11.5

1. Determine whether the plane through the three points

(100,−200, 300), (150, 300, 200), (125, 150, 100),

is parallel to the plane through the three points

(121, 176, 2309), (250, 376, 3686), (300,−250, 5275).

If the two planes are parallel, then find the distance between them. If the two planes are not parallel then
find parametric equations for their line of intersection.

2. Find the distance between the point (125, 265, 755) and the plane

26x − 3y − 2z + 2000 = 0.

3. Find the angle between the normal to the plane

26x − 3y − 2z + 2000 = 0

and the vector
< 75, 37, 47 > .

